

Chapter 4

Differentiation

We start this chapter by computing derivatives using the limit definition of the derivative. Then we introduce the Maple syntax for differentiation. Maple makes a distinction between expressions and functions (See Chapter 2.), and we discuss the syntax for differentiating both. We conclude this chapter with sections on implicit differentiation and linear approximation.

4.1 The Limit of the Difference Quotient

The definition of the derivative of the function $f(x)$ at the point $x = a$ is given by

$$f'(a) = \lim_{h \rightarrow 0} \frac{f(a+h) - f(a)}{h}$$

The motivation of this definition is that the derivative of f at $x = a$ should be the slope of the tangent line to the graph of f at $x = a$. The key idea is that the quantity $\frac{f(a+h) - f(a)}{h}$ is the slope of the line segment connecting the points $(a, f(a))$ and $(a+h, f(a+h))$. As h approaches zero, the point $(a+h, f(a+h))$ approaches $(a, f(a))$, the line segment rotates into the tangent line and the slope of the line segment approaches the slope of the tangent.

EXAMPLE: Consider the function $f(x) = x^3 - 8x$. Compute $f'(2)$ and the equation of the tangent line. Then plot the function and its tangent line in the same plot.

SOLUTION: Enter `f` as a Maple function and calculate the above limit with $a = 2$.

```
> f:=x->x^3-8*x;
```

$$f := x \rightarrow x^3 - 8x$$

```
> Limit((f(2+h)-f(2))/h,h=0); m:=value(%);
```

$$\lim_{h \rightarrow 0} \frac{(2+h)^3 - 8 - 8h}{h}$$

$$m := 4$$

Note that the `Limit` command displays the limit so that you can check for typing errors and then the `value(%)` command evaluates this limit (recall that the percent `%` refers to the result of the previous command, which in this case is the limit).


So the slope of the tangent line at the point $(2, f(2)) = (2, -8)$ is $m = 4$. Accordingly, its formula is

$$> y := m * (x - 2) + f(2);$$

$$y := 4x - 16$$


This tangent line and the function can be plotted with two colors using

```
> plot([y, f(x)], x=0..4, color=[green, blue]);
```



To see that this tangent line closely approximates the graph of the function f near $x = 2$, replot this graph over a small interval about $x = 2$.

```
> plot([y, f(x)], x=1.9..2.1, color=[green, blue]);
```



Another way to obtain $f'(2)$ is to compute $f'(x)$ and then substitute $x = 2$. To do this, compute the above limit with $a = x$ and store the result in the variable named `Df`.

```
> Limit((f(x+h)-f(x))/h,h=0); Df:=value(%);
```

$$\lim_{h \rightarrow 0} \frac{(x+h)^3 - 8h - x^3}{h}$$

$$Df := 3x^2 - 8$$

To obtain $f'(2)$, evaluate Df at $x = 2$:

```
> eval(Df, x=2);
```

4

which is the same slope as above.

Let's examine the above limit process more carefully. First, enter the difference quotient $\frac{f(x+h) - f(x)}{h}$ into Maple. Store this expression as the variable `diffq` and `simplify`.

```
> diffq:=(f(x+h)-f(x))/h;
```

$$diffq := \frac{(x+h)^3 - 8h - x^3}{h}$$

```
> simplify(diffq);
```

$$3x^2 + 3xh + h^2 - 8$$

Notice that the derivative, $3x^2 - 8$, is obtained by letting h tend to zero (any term containing an h will disappear).

4.2 Differentiating Functions

All the rules for differentiation are programmed into Maple, making it easy to differentiate complicated functions. The Maple syntax for differentiating functions is different from that for expressions. So read carefully!

Suppose that `f` is defined as a Maple function. Then `D(f)` is the *function* that represents the derivative of `f`.

EXAMPLE 1: Enter $f(x) = x^2(x^5 + 1)$ as a function and compute $f'(2)$.

SOLUTION: The function is

```
> f:=x->x^2*(x^5+1);
```

$$f := x \rightarrow x^2(x^5 + 1)$$

and the derivative is

```
> D(f);
```

$$x \rightarrow 2x(x^5 + 1) + 5x^6$$

The arrow notation indicates that `D(f)` is a function. To evaluate $f'(2)$, type

```
> D(f)(2);
```

452

If you are going to use the derivative repeatedly, it is best to give it a name. Thus in

```
> Df:=D(f);
```

$$Df := x \rightarrow 2x(x^5 + 1) + 5x^6$$

$D(f)$ is the operation of taking the derivative of f , while Df is the name for the result. Then the derivatives of f at the points $x = -3$, $x = 3$, and $x = t$ are obtained by typing

```
> Df(-3), Df(3), Df(t);
```

$$5097, 5109, 2t(t^5 + 1) + 5t^6$$

EXAMPLE 2: Find the tangent line at $x = 2$ for the function

```
> f:=x->(x^3-1)/(x+2);
```

$$f := x \rightarrow \frac{x^3 - 1}{x + 2}$$

Then plot the function and its tangent line.

SOLUTION: The tangent line passes through the point $(2, f(2))$ and has a slope $f'(2)$, which in Maple is $D(f)(2)$. First assign the slope to the variable m .

```
> m:=D(f)(2);
```

$$m := \frac{41}{16}$$


Then the tangent line is given by

```
> y:=m*(x-2)+f(2);
```

$$y := \frac{41x}{16} - \frac{27}{8}$$

A plot of the expressions $f(x)$ and y shows the tangent line has been found.

```
> plot([f(x),y],x=0..4);
```



Higher derivatives can be evaluated by repeated application of the D operator. For example, $D(D(f))$ represents the second derivative of the function f (so $f''(2)$ can be computed by entering $D(D(f))(2)$). Alternatively, the syntax $(D@@2)(f)$ also represents the second derivative. This syntax is preferred for higher derivatives; e.g., the 8th derivative, $f^{(8)}(2)$ is evaluated by typing

```
> (D@@8)(f)(2);
```

$$\frac{-2835}{2048}$$

4.3 Differentiating Expressions

The `D` operator only works for functions. Sometimes, however, it is more convenient to work with expressions. To differentiate an expression, you first display the derivative using the `Diff` command. The first argument is the expression to be differentiated while the second argument is the variable of differentiation. This is followed by `value(%)`; to compute the derivative.

EXAMPLE 1: Find the derivative of $f = \frac{3x^2+4x^3}{2x+1}$ and evaluate it at $x = 2$.

SOLUTION: The derivative is

```
> Diff((3*x^2+4*x^3)/(2*x+1),x); Df:=value(%);
```

$$Df := \frac{d}{dx} \left(\frac{3x^2 + 4x^3}{2x + 1} \right) = \frac{6x + 12x^2}{2x + 1} - \frac{2(3x^2 + 4x^3)}{(2x + 1)^2}$$

By first displaying the derivative, you can be sure it is entered correctly before differentiating. Notice that the derivative of `f` has been assigned to the variable `Df`. If you want to simplify the answer, execute

```
> Df:=simplify(Df);
```

$$Df := \frac{2x(8x^2 + 9x + 3)}{(2x + 1)^2}$$

To evaluate the derivative at $x = 2$, use the `subs` or `eval` command.

```
> subs(x=2, Df);
```

$$\frac{212}{25}$$

If the expression f has already been assigned to a label `f`, as is frequently the case, then it is more convenient to differentiate directly using the `diff` command.

EXAMPLE 2: Enter the function $\cos^3(t^2)$ as an expression and find its derivative at $t = \sqrt{\pi}/2$.

SOLUTION: In Maple the power on a trig function, such as the cube on $\cos^3(t^2)$, must be entered after the argument as follows:

```
> f:=cos(t^2)^3;
```

$$f := \cos(t^2)^3$$

Now differentiate f with respect to t and give it a label.

```
> df:=diff(f,t);
```

$$df := -6 \cos(t^2)^2 \sin(t^2) t$$

Finally evaluate at $t = \sqrt{\pi}/2$:

```
> eval(df, t=sqrt(Pi)/2);
```

$$-\frac{3\sqrt{2}\sqrt{\pi}}{4}$$

NOTE: `eval` automatically simplifies. `subs` would not. Try it!

Higher derivatives can be calculated by repeating the variable of differentiation or by using a dollar sign followed by the number of derivatives to be taken. For example, the second and third derivatives of f are

```
> d2f:=simplify(diff(f,t,t)); d3f:=(diff(f,t$3));
      d2f := -6 cos(t^2) (-4 t^2 + 6 cos(t^2)^2 t^2 + cos(t^2) sin(t^2))
d3f := -48 sin(t^2)^3 t^3 + 168 cos(t^2)^2 sin(t^2) t^3 + 72 cos(t^2) sin(t^2)^2 t - 36 cos(t^2)^3 t
```

If f is a function, its derivative is either the function $D(f)$ or the expression $\text{diff}(f(x),x)$. (Here, you must type $f(x)$ and not just f .) The disadvantage in the `diff` syntax is that you must use a more cumbersome `subs` or `eval` command to evaluate the derivative at particular points.


4.4 Implicit Differentiation

Up until now, we have discussed functions that are defined *explicitly*, which means that the dependent variable, such as y or f , is isolated on one side of an equation and an expression involving only the dependent variable, say x , appears on the other side (e.g., $y = x^2$). In this chapter, we consider functions that are given *implicitly*. This means that the variables are often mixed together in an equation.

An example is the Folium of Descartes, which is given by the equation $3xy = x^3 + y^3$. Note that it would be very difficult to solve for the variable y explicitly in terms of x since this is a cubic in y . Nevertheless, plots and derivatives can still be obtained.


To plot the equation, $3xy = x^3 + y^3$, use the `implicitplot` command from the `plots` package.

```
> with(plots):
> eq:=3*x*y=x^3+y^3;
      eq := 3 x y = x^3 + y^3
> implicitplot(eq, x=-3..3, y=-3..3, scaling=constrained);
```



You can increase the number of points plotted by adding the option `grid=[p,q]`, where `p` and `q` are the numbers of points in the x and y directions. The default is `grid=[25,25]`.

```
> implicitplot(eq, x=-3..3, y=-3..3, scaling=constrained,
> grid=[201,201]);
```



Note that this plot contains a loop, and cannot be described globally as the graph of one function $y = y(x)$. However, near most points a piece of the plot is the graph of one function. For example, the lower piece of the loop over the interval $-1 \leq x \leq 1$ is the graph of one function $y(x)$. Finding a formula for $y(x)$ involves solving the equation $3xy = x^3 + y^3$ for y in terms of x . This is difficult since this equation involves a cubic. However, it is possible to find numerical values of $y(x)$ at specific values of x using `fsolve`. For example, the three values of y at $x = 1$ are found using:


```
> eq1:=subs(x=1, eq);
      eq1 := 3y = 1 + y^3
> fsolve(eq1,y);
      -1.879385242, 0.3472963553, 1.532088886
```

What about the points where $y = x$? They can be found exactly:

```
> solve({y=x,eq},{x,y});
      {x = 0, y = 0}, {x = 0, y = 0}, {x = 3/2, y = 3/2}
```


Near the point $(x, y) = (1.5, 1.5)$ the graph is actually a function, as can be seen from an enlarged plot that limits the range of x and y .

```
> implicitplot(eq, x=1.25..1.75, y=1.25..1.75, scaling=constrained);
```



Indeed, over a very small plot range, the graph usually looks like a straight line (the tangent line).

```
> implicitplot(eq, x=1.49..1.51, y=1.49..1.51, scaling=constrained);
```



Implicit differentiation is the procedure used to find the derivative of an implicitly defined function. We will demonstrate the sequence of Maple commands used to find implicit derivatives on the Folium of Descartes.

First, assign the label `eq` to the equation. Then replace y by $y(x)$ (the unknown function).

```
> eq:=3*x*y=x^3+y^3; eq1:=subs(y=y(x),eq);
```

$$eq := 3xy = x^3 + y^3$$

$$eq1 := 3xy(x) = x^3 + y(x)^3$$

Next, take the derivative of both sides of the equation. Each side of this equation is an expression in x , so the `diff` command must be used to differentiate it. The left side requires the product rule; the right side requires the chain rule.

```
> Deq:=diff(eq1,x);
```

$$Deq := 3y(x) + 3x \left(\frac{d}{dx} y(x) \right) = 3x^2 + 3y(x)^2 \left(\frac{d}{dx} y(x) \right)$$

Then solve for the derivative $\frac{d}{dx}y(x)$ and label it `Dy`:

```
> Dy:=solve(Deq,diff(y(x),x));
```

$$Dy := \frac{y(x) - x^2}{-x + y(x)^2}$$

There is no further need to emphasize that y depends on x ; so substitute y back for $y(x)$: (Simplify if necessary.)

```
> Dy:=eval(Dy,y(x)=y);
```

$$Dy := \frac{y - x^2}{-x + y^2}$$

The expression Dy is the *implicit derivative*, and its numerical value can be determined at any point (x, y) by inserting the x and y values into its formula. For instance, the slope at the point $(3/2, 3/2)$ is

```
> m:=subs({x=3/2,y=3/2},Dy);
```

$$m := -1$$

This result agrees with the preceding graph.

4.5 Linear Approximation

In the previous section, we studied the Folium of Descartes,

```
> eq:=3*x*y=x^3+y^3;
```

$$eq := 3xy = x^3 + y^3$$

which implicitly defines a function $y = f(x)$ in the neighborhood of the point $\left(\frac{3}{2}, \frac{3}{2}\right)$. We found the slope at this point is $f' \left(\frac{3}{2}\right) = -1$.

```
> (a,b):=(3/2,3/2); m:=-1;
```

$$a, b := \frac{3}{2}, \frac{3}{2}$$

$$m := -1$$

So the tangent line is

```
> taneq:=y=m*(x-a)+b;
```

$$taneq := y = -x + 3$$


We can convert the right hand side of this equation into a function by using the `rhs` and `unapply` commands:

```
> ftan:=unapply(rhs(taneq),x);
```

$$ftan := x \rightarrow -x + 3$$

This tangent line function represents the linear approximation to the graph of the original function, i.e. the linear function $y = f_{tan}(x)$ approximates the values of $y = f(x)$ for x near 1.5. To plot both the curve and its tangent line, use the `implicitplot` command.

```
> with(plots):
> implicitplot([eq,taneq], x=1..2, y=1..2, scaling=constrained);
```



Let's compare the values of $y = f(x)$ and $y = f_{tan}(x)$ at $x = 1.51$. To find the value of $f(x)$ at $x = 1.51$, use `subs` and `fsolve`.

```
> subs(x=1.51,eq); fsolve(%,y,y=1.4..1.6);
4.53 y = 3.442951 + y^3
1.489436032
```

On the other hand, the value from the linear approximation is

```
> ftan(1.51);
1.49
```

These are very close.

4.6 Summary

Differentiation of Functions and Expressions

- Compute a derivative (i.e., the slope of a tangent line) by finding the limit of the difference quotient. Here are the steps:

```
f:=x-> expr; ..... Define the function.
Limit((f(a+h)-f(a))/h,h=0); ... Compute the limit of the
m:=value(%); ..... difference quotient.
```

- Compute the line tangent to the graph of $y = f(x)$. Display both the curve and its tangent line on the same graph.

```
y:=m*(x-a)+f(a); ..... Define the tangent line.
plot([y,f(x)],x=0..4); ... Plot the curve and the tangent line.
```

- Compute the derivative of a Maple function f , and evaluate it.

```
Df:=D(f) ..... Compute f'(x).
Df(a); ..... Evaluate f'(a).
```

- Compute a higher derivative of a Maple function f , and evaluate it.


```
Dnf:=(D@@n)(f) ..... Compute  $f^{(n)}(x)$ .
Dnf(a); ..... Evaluate  $f^{(n)}(a)$ .
```
- Compute the derivative of a Maple expression f , and evaluate it.


```
Df:=diff(f,x) ..... Compute  $f'(x)$ .
subs(x=a,Df); ..... Evaluate  $f'(a)$ .
```
- Compute a higher derivative of a Maple expression f , and evaluate it.


```
Dnf:=diff(f,x$n) ..... Compute  $f^{(n)}(x)$ .
subs(x=a,Dnf); ..... Evaluate  $f^{(n)}(a)$ .
```

Implicit Differentiation

- Maple expressions and functions explicitly define functions, $y = f(x)$.
- Maple equations implicitly define functions, $y = f(x)$, along pieces of their graphs. One equation may define several implicit functions.
- Plot a Maple equation.


```
with(plots); ..... Load the plots package.
eq:=equation; ..... Define the equation.
implicitplot(eq, x=p..q, y=r..s); ... Plot the equation.
```
- The implicit derivative is the derivative of an implicit function. Maple can find this derivative, even if it cannot solve for y .
- Compute an implicit derivative and evaluate at a point.


```
eq1:=subs(y=y(x),eq); ..... Express  $y$  as a function of  $x$ .
Deq:=diff(eq1,x); ..... Differentiate the equation.
Dy:=solve(Deq,diff(y(x),x)); Solve for the derivative.
Dy:=subs(y(x)=y,Dy); ..... Suppress the dependence on  $x$ .
Dy:=simplify(Dy); ..... Simplify if necessary.
m:=subs(x=a,y=b,Dy); ..... Evaluate at a point.
```

Linear Approximation

- The linear approximation to a curve at a point $(x, y) = (a, b)$ on the curve is the function given by the formula for the tangent line.


```
taneq:=y=m*(x-a)+b; ..... Equation of the tangent line.
ftan:=unapply(rhs(taneq),x); ... Convert equation to function.
with(plots); ..... Load the plots package.
implicitplot({eq,taneq},
x=p..q, y=r..s); ..... Plot the function and tangent.
```

4.7 Exercises

1. Define each of the following as a Maple function. Find the slope of the secant line between the two points. NOTE: θ is entered as **theta**.

(a) $f(x) = \frac{x^2 - x}{x^2 + 1}$ between $x = 2$ and $x = 4$

(b) $f(\theta) = \cos(\theta)$ between $\theta = \pi/6$ and $\theta = \pi/3$

(c) $f(x) = 3^x$ between $x = 0$ and $x = 2$

2. Define each of the following as a Maple function. Find the slope of the secant line between $x = a$ and $x = a + h$. Simplify the answer using **expand**, **factor** and/or **simplify**. Find the slope of the tangent line by taking the limit of the slope of the secant line as h approaches 0.

(a) $f(x) = \frac{x^2 - x}{x^2 + 1}$ for $a = 2$

(b) $f(\theta) = \cos(\theta)$ for $a = \pi/6$

(c) $f(x) = 3^x$ for $a = 0$

3. Define each of the following as a Maple function. Find the derivative using **D** and evaluate at a .

(a) $f(x) = \frac{x^2 - x}{x^2 + 1}$ for $a = 2$

(b) $g(t) = \cos^2(t^3 + 1)$ for $a = 1$

(c) $r(\theta) = \theta^2 + \sin(\theta) \cos(\theta)$ for $a = \pi/3$

4. Repeat Exercise 3 but define the given functions as expressions and then use **diff** to take the derivatives.
5. Define $f = \sin(px^3) + \cos(qx^2)$ as a function of x (with p and q constant) and differentiate with respect to x . Now define f as a function of q (with x and p constant) and differentiate with respect to q .
6. Repeat Exercise 5 but define the function as an expression and use **diff** to take the derivative.
7. Define $f(x) = 2x^4 - 3x^2$ as a Maple function. Find the equation of the tangent line at the point $x = 2.3$. Plot both the function and the tangent line on the same coordinate axes, with different colors.
8. Repeat Exercise 7 but define the function as an expression.
9. Find the (angle of) inclination of the tangent line in Exercise 7 in degrees. HINT: Recall that the tangent of the inclination is the slope of the line. So this problem amounts to taking the inverse tangent of the slope of the line in Exercise 7. In Maple, the inverse tangent, **arctan**, gives radians.

10. Plot the line $y = 1 - 3x$ and the curve $y = \sqrt{34.6 - x^2}$. (In Maple, enter the \sqrt{u} as `sqrt(u)`.) Find their point of intersection. Then find the acute angle between the tangent lines at this point of intersection. Look at the plot to determine if you need to add or subtract their inclinations.
11. Define $f(x) = \sin(x) + 2\cos(x)$ and $g(x) = x^3 - 2x + 4$ as Maple functions. Then use `D` to differentiate $f(x)g(x)$, $f(x)/g(x)$, $(f \circ g)(x)$ and $(g \circ f)(x)$. (For Maple functions, the product fg is entered as `f*g`, the quotient f/g is entered as `f/g` and the composition $f \circ g$ is entered as `f@g`.) Evaluate these functions at x to get the expressions.
NOTE: The derivatives written as functions are very cryptic.
12. For each equation, use implicit differentiation to find the slope at the indicated point. Then plot the equation and the tangent line at the indicated point.
- (a) $x^4 + x^2y^2 + y^4 = 48$ at $(x, y) = (2, 2)$
 (b) $(x^2 + y^2 - 2x)^2 = 16x^2 + 12y^2$ at $(x, y) = (2, 4)$
 (c) $x^2y + xy^2 = 30$ at $(x, y) = (3, 2)$
13. For each equation, use `implicitplot` to plot the given equation. Then find the y -values of the upper piece of its plot at $x = 1, 1.25, 1.5, 1.75$ and 2 . Compute the slopes of the tangent lines at $x = 1$ and $x = 1.25$ by implicit differentiation and compare these values to the slope of the secant between $x = 1$ and $x = 1.25$.
- (a) $(x^2 + y^2)^2 = 9(x^2 - y^2)$
 (b) $x^2y + xy^2 = 25$
14. Graph the equations $y = x^2$ and $x^4 - 3x + 2y^2 + 6y = 14$. Be sure to use `scaling=constrained`. Find the point(s) of intersection. For each point of intersection, find the acute angle between the tangent lines of both equations.
15. By following the steps given below, show that the derivative obtained implicitly (without solving for y) is the same as the derivative obtained explicitly by solving for y .
- (a) Consider the equation $x^2 + xy + y^2 = 81$. Use the procedure described in the text to compute the implicit derivative and label it `Dy`.
 (b) Use Maple to solve the equation for y and label the result as `sol`.
 (c) Execute `Dy1:=subs(y=sol[1],Dy)`; to replace y by its explicit form.
 (d) Execute `Dy2:=diff(sol[1],x)`; to take the derivative explicitly.
 (e) Compare your answers in (c) and (d) by subtracting.
 (f) Repeat your work with `sol[2]`.


16. One of the virtues of the implicit derivative process is that, given an equation relating x and y , it is not necessary to explicitly solve for y in order to compute y' . On the other hand, the answer is sometimes not as nice as an explicit answer.
- Enter the equation $3y^2 - 7xy + 2x^2 = 0$ and label it `eq`. Use the procedure described in the text to compute the implicit derivative.
 - Now use `with(plots):` and `implicitplot` to have Maple draw the graph. (Increase the `grid` size.) Give a better answer for the derivative than the one you got in (a).
 - Try `factor(eq)`; . Now explain why the answer in (b) is true.
 - Without executing any Maple commands, use your *insight* from the graph and `factor` to give the derivative at any point on the graph, except the origin. What is the derivative at the origin?
17. In the case of the Folium of Descartes, $3xy = x^3 + y^3$, for *most* points (x, y) on the graph there is a plot range for which the graph is a function. However, there are two exceptions where there are vertical tangents.
- After plotting with the `implicitplot` command, click with your mouse to find approximate floating point decimal coordinates for the two points on the graph that have vertical tangents.
 - Refine your guess as to the coordinates of these two points by noting that the denominator of y' is $x - y^2$. The tangent is vertical when $x - y^2 = 0$. Use `fsolve` to find a better approximation. You may need to limit the range of the x - and y -variables as in:


```
> fsolve({x-y^2=0,eq}, {x=p..q,y=r..s});
```
 - To find the *exact* x - and y -coordinates of the two points, use


```
> solve({x-y^2=0,eq},{x,y});
```

 The answer is given in terms of `RootOf`'s, as discussed in Section 2.3. Resolve the `RootOf`'s using `allvalues`. Confirm the real solutions agree with your approximate solutions by applying `evalf` to the exact answers.
18. An object's position at time t is $x(t) = 481t - 183t^2 + 24t^3 - t^4$. Find all times T when the instantaneous velocity of the object (its derivative) equals the average velocity of the object over the interval $0 \leq t \leq T$. (There are 3 solutions, but ignore the one at $T = 0$.) Then plot $x(t)$ and the two tangent lines on one plot. (NOTE: The tangent line at $T = a$ is `tan1:=x(a)+Dx(a)*(t-a);`.) What do you observe and why does it happen?
19. *This exercise does not involve Maple. It will confirm your understanding of the answer in the previous problem and provide a warm-up for the next exercise, which does involve Maple.* Suppose the curve pictured here

represents a plot of an object's distance versus time. From this plot, determine the approximate time T when the instantaneous velocity of the object equals the average velocity of the object over the interval $0 \leq t \leq T$.




20. Consider the following data, which represent the position (in meters) of an object at various times (in seconds).

Time	1	2	3	4	5	6	7	8	9	10
Pos.	0.80	1.34	1.72	2.06	2.21	2.49	2.81	3.41	4.19	4.64

Plot this data set using Maple. From this plot, estimate the time(s) T at which the instantaneous velocity is equal to the average velocity over the time interval $0 \leq t \leq T$.

HINT: Enter the data as a list, with alternating times and positions, i.e., `data=[[1,0.80],[2,1.24], ...]`; as done in Section 3.4. Then type `plot(data);`. This will connect all the data points with line segments. To obtain a plot of data points without the connecting line segments, add the option `style=point`.

21. *Newton's Method.* We have already seen examples of solving equations, such as $x^3 + x - 1 = 0$, numerically with the `fsolve` command. But how does Maple do it? The point of this exercise is to explore one algorithm, called Newton's method, which is often used to solve equations numerically. The basic idea behind Newton's method is as follows: (For more details, see the section on Newton's method in your text.) To solve the equation $f(x) = 0$, pick a starting point, x_0 , near the solution to $f(x) = 0$ (for example, from a graph, choose the closest integer to the solution). Construct the tangent line to $y = f(x)$ at $x = x_0$. Generally, the x -intercept of the tangent line is closer to the solution than is x_0 .



We denote this x -intercept by x_1 . After computing the equation of the tangent line and finding its x -intercept, we obtain

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}.$$

Then we iterate this process. This leads to a sequence of points

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)}$$

which approaches the solution to the equation $f(x) = 0$ as n gets bigger (in fact, n usually does not have to get too large). This algorithm can be made into a Maple command as follows:

```
> Df:=D(f);
> Newton:=x->evalf(x-f(x)/Df(x));
```

If x is taken as x_n then `Newton(x)` will be x_{n+1} . This command can then be used to solve $x^3 + x - 3 = 0$ using one of the following methods:

- i. Enter the function and compute the derivative:

```
> f:=x->x^3+x-3; Df:=D(f);
```

Plot it and notice there is a solution of $f(x) = 0$ in the interval $1 \leq x \leq 2$. So start with $x_0 = 1$:

```
> x0:=1;
```

To get x_1 , the first iteration of Newton's method, execute


```
> Newton(%);
```

Now repeatedly execute the command `Newton(%)`; until the output no longer changes. (Use copy and paste to obtain multiple copies of the command.) Compare your answers with the result of Maple's `fsolve` command.

- ii. You can also write a loop to execute Newton's method, say 5 times:

```
> x0:=1;
> for n from 1 to 5 do
> x[n]:=Newton(%)
> end do;
```

The commands between the `do` and the `end do;` will be executed 5 times (with the variable n as the counter). The value of x after this program is executed will be an approximation to the solution. Compare the loop with your result in part (a).

- iii. The loop can be automated to execute Newton's method until the difference between successive approximations (i.e., $|x_{n+1} - x_n|$) is less than some pre-assigned small number, such as 10^{-12} . One way to do this is to keep track of the current value `xnew` and previous value `xold` and assign the variable `tol` (for tolerance) to the absolute value of their difference. The algorithm should continue to execute until `tol` is less than the pre-assigned value. To implement this in Maple, first define the desired tolerance and set the current tolerance to 1 so the loop can start. Also increase the number of `Digits`.

```
> desiretol:=10^(-12); tol:=1; Digits:=15;
```

Then the loop is

```
> xold:=1;
> for n from 1 to 100 while (tol>desiretol) do
> xnew:=Newton(xold);
> tol:=abs(xnew-xold);
> xold:=xnew;
> end do;
```

This syntax will execute the statements between the `do` and the `end do` as long as the variable `tol` exceeds 10^{-12} or until the counter n reaches 100. (The choice of 100 is arbitrary but is included to keep the program from running indefinitely if `tol` never gets below 10^{-12} .)


For this exercise, use Newton's method to find all solutions to each of the following equations to 25 decimal places. (Remember to set your `Digits` back to 10 afterwards.) Compare your answers with the result of `fsolve`.

(a) $x^3 - 13x - 7 = 0$ (3 solutions)

(b) $x^3 = 3x - 1.5$ (3 solutions)

(c) $4 \cos(x) = 0.9x$ (3 solutions)

22. A cylindrical can is to contain 800 cubic centimeters. Find the dimensions of the can if its surface area including the top and bottom is 500 square centimeters. HINT: Plot the equation you need to solve. Notice there are two answers. Use several iterations of Newton's method to solve the resulting equation near each solution. Compare your answers to the ones obtained by using `fsolve`.
23. An waffle cone is to contain 51 cubic centimeters of ice cream. Find the dimensions (radius and height) of the cone if its lateral surface area is 71 square centimeters. Use Newton's method to solve the equation. Compare your answers to the ones obtained by using `fsolve`. Notice there are two solutions. Which one is real world? NOTE: The lateral surface area of a cone is $A = \pi r \sqrt{r^2 + h^2}$
24. The bottom of a trough is constructed from a 4 foot by 6 foot rectangular piece of metal by bending it so that the 4 foot width forms an arc of a circle. (See the figure)



If the volume of the trough is 5 cubic feet, find the angle θ subtended by the arc. Use several iterations of Newton's method to solve the resulting equation for the volume (see the hint below) and compare your answer to the one obtained by using `fsolve`.


HINT: Suppose that r is the radius of the circular arc subtended by the angle θ . Explain why the length of the arc is $L = r\theta = 4$ and why the area of an end of the trough is

$$A = \frac{r^2\theta}{2} - \frac{r^2 \sin(\theta)}{2}$$

(Think of the area of the end as the area of the circular sector minus the area of a triangle.) Use these equations to show the volume of the trough is:

$$V = \frac{48(\theta - \sin(\theta))}{\theta^2}$$

25. A pulley consists of an 40 inch band tautly wrapped around two wheels of radius 3 inch and 4 inches, respectively, as shown in the diagram. Find the length of the straight pieces of the band that are not in contact with either wheel. Use several iterations of Newton's method to solve the equation and compare your answer to the one obtained by using `fsolve`.



HINT: Let c be the distance between the centers and let θ be the angle between the line between their centers and a radius to the point where the band makes contact with the wheel. The two unknowns are θ and c . The two equations express (i) the total length of the band and (ii) the fact that the radii to the points of contact are perpendicular to the piece of band not in contact. Express the points of contact in terms of θ and c . Use the facts that the length of an arc is $l = r\theta$ and two lines are perpendicular if their slopes are negative reciprocals. Solve for c exactly from one equation and substitute into the other.